Computergestuetzte Mathematik zur Analysis

Lektion 14 (4. Februar)

```
> restart:
```

Gewöhnliche Differentialgleichungen II

```
> os := diff(y(x),x$2) + y(x);
  #harmonischer Oszillator
```

$$os := \frac{d^2}{dx^2}\, y(x) + y(x) \qquad \textbf{(1.1)}$$

```
> dsolve(os=0,y(x));
```

$$y(x) = \_C1 \sin(x) + \_C2 \cos(x) \qquad \textbf{(1.2)}$$

```
> dsolve({os=0,y(0)=1,D(y)(0)=0},y(x));
```

$$y(x) = \cos(x) \qquad \textbf{(1.3)}$$

```
> l1 := rhs((1.3));
```

$$l1 := \cos(x) \qquad \textbf{(1.4)}$$

```
> gos:= diff(y(x),x$2) + 1/5*diff(y(x),x) + y(x);
  #harmonischer Oszillator mit Daempfung
```

$$gos := \frac{d^2}{dx^2}\, y(x) + \frac{1}{5}\, \frac{d}{dx}\, y(x) + y(x) \qquad \textbf{(1.5)}$$

```
> dsolve({gos=0,y(0)=1,D(y)(0)=0},y(x));
```

$$y(x) = \frac{1}{33}\sqrt{11}\, e^{-\frac{1}{10}x} \sin\left(\frac{3}{10}\sqrt{11}\, x\right) + e^{-\frac{1}{10}x} \cos\left(\frac{3}{10}\sqrt{11}\, x\right) \qquad \textbf{(1.6)}$$

```
> l2 := rhs((1.6));
```

$$l2 := \frac{1}{33}\sqrt{11}\, e^{-\frac{1}{10}x} \sin\left(\frac{3}{10}\sqrt{11}\, x\right) + e^{-\frac{1}{10}x} \cos\left(\frac{3}{10}\sqrt{11}\, x\right) \qquad \textbf{(1.7)}$$

```
> plot([l1,l2],x=0..20);
```



Inhomogene Gewöhnliche Differentialgleichungen

```
> l3:=rhs(dsolve({os=sin(x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l3 := \cos(x) + \frac{1}{2}\sin(x) - \frac{1}{2}\cos(x)\,x \tag{2.1}$$

```
> plot(l3,x=0..100); # Resonanzfall
```




```
> l4:= rhs(dsolve({os=sin(3/4*x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l4 := -\frac{12}{7}\sin(x) + \cos(x) + \frac{16}{7}\sin\left(\frac{3}{4}x\right) \quad \textbf{(2.2)}$$

```
> l5:= rhs(dsolve({os=sin(7/8*x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l5 := -\frac{56}{15}\sin(x) + \cos(x) + \frac{64}{15}\sin\left(\frac{7}{8}x\right) \quad \textbf{(2.3)}$$

```
> plot([l4,l5],x=0..100,color=[red,blue]);
```



▶ Schwingende Membran (Bessel Funktionen)

▼ Differentialgleichungssysteme

```
> restart:
> with(LinearAlgebra):
> A:=<<0|1|0>,<-1|0|1>,<0|0|2>>;
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 1 \\ 0 & 0 & 2 \end{bmatrix} \tag{4.1}$$

```
> T:=MatrixExponential(A,t);
```

(4.2)

$$T := \begin{bmatrix} \cos(t) & \sin(t) & -\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}e^{2t} \\ -\sin(t) & \cos(t) & -\frac{2}{5}\cos(t) + \frac{2}{5}e^{2t} + \frac{1}{5}\sin(t) \\ 0 & 0 & e^{2t} \end{bmatrix} \quad (4.2)$$

```
> #Loesung y' = A*y , y(0) = <a,b,c>
> y0 := <a,b,c>;
```

$$y0 := \begin{bmatrix} a \\ b \\ c \end{bmatrix} \quad (4.3)$$

```
> y(t) := T.y0;
```

$$y(t) := \begin{bmatrix} \cos(t)\, a + \sin(t)\, b + \left(-\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}e^{2t}\right) c \\ -\sin(t)\, a + \cos(t)\, b + \left(-\frac{2}{5}\cos(t) + \frac{2}{5}e^{2t} + \frac{1}{5}\sin(t)\right) c \\ e^{2t} c \end{bmatrix} \quad (4.4)$$

```
> with(VectorCalculus):
> BasisFormat(false):
> diff(y(t),t) - A.y(t);
```

$$\left[\left[0\right], \left[\left(\frac{2}{5}\sin(t) + \frac{4}{5}e^{2t} + \frac{1}{5}\cos(t)\right) c + \left(-\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}e^{2t}\right) c - e^{2t} c\right], \left[0\right]\right] \quad (4.5)$$

```
> simplify((4.5))
```

Warning, inserted missing semicolon at end of statement

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \quad (4.6)$$

```
> eval(y(t),t=0);
```

$$\begin{bmatrix} a \\ b \\ c \end{bmatrix} \quad (4.7)$$

Das Pendel

```
> restart;
> Dgl := diff(y(t),t$2)= -sin(y(t));
```

$$Dgl := \frac{d^2}{dt^2} y(t) = -\sin(y(t)) \quad (5.1)$$

```
> AW:= y(0)=Pi/8,D(y)(0)=0;
```

$$AW := y(0) = \frac{1}{8}\pi, D(y)(0) = 0 \quad (5.2)$$

```
> dsolve({Dgl,AW},y(t));
```

$$y(t) = RootOf\left(\int_{\_Z}^{\frac{1}{8}\pi} \frac{1}{\sqrt{2\cos(\_a) - 2\cos\left(\frac{1}{8}\pi\right)}} d\_a + t\right), y(t) = RootOf\left(\int_{\frac{1}{8}\pi}^{\_Z} \frac{1}{\sqrt{2\cos(\_a) - 2\cos\left(\frac{1}{8}\pi\right)}} d\_a + t\right) \quad (5.3)$$

```
> Lsg:=dsolve({Dgl,AW},y(t),type=numeric,output=listprocedure);
```

$Lsg := \left[t = \textbf{proc}(t) \ \ldots \ \textbf{end proc}, y(t) = \textbf{proc}(t) \ \ldots \ \textbf{end proc}, \frac{d}{dt} y(t) = \right.$ (5.4)

proc(t)

...

end proc]

```
> yl := eval(y(t),Lsg);
```

$$yl := \textbf{proc}(t) \ \ldots \ \textbf{end proc} \quad (5.5)$$

```
> yl(1);
```

0.215837134280979 (5.6)

```
> Dgl_os := diff(y(t),t$2) = -y(t);
```

$$Dgl\_os := \frac{d^2}{dt^2} y(t) = -y(t) \quad (5.7)$$

```
> dsolve({Dgl_os,AW},y(t));
```

$$y(t) = \frac{1}{8}\pi\cos(t) \quad (5.8)$$

```
> yl_os:=unapply(rhs((5.8)),t);
```

$$yl\_os := t \rightarrow \frac{1}{8}\pi\cos(t) \quad (5.9)$$

```
> plot([yl,yl_os],0..50,color=[black,red]);
```




```
> AW2:= y(0)=Pi/4,D(y)(0)=0;
```

$$AW2 := y(0) = \frac{1}{4}\pi, \mathrm{D}(y)(0) = 0 \quad \textbf{(5.10)}$$

```
> Lsg:=dsolve({Dgl,AW2},y(t),type=numeric,output=listprocedure);
```

$Lsg :=$ [$t = $ **proc**(t) ... **end proc**, $y(t) = $ **proc**(t) ... **end proc**, $\frac{\mathrm{d}}{\mathrm{d}t}\, y(t) = $ **proc**(t) ... **end proc**] **(5.11)**

```
> yl := eval(y(t),Lsg);
```

$yl :=$ **proc**(t) ... **end proc** **(5.12)**

```
> dsolve({Dgl_os,AW2},y(t));
```

$$y(t) = \frac{1}{4}\pi\cos(t) \quad \textbf{(5.13)}$$

```
> yl_os:=unapply(rhs((5.13)),t);
```

$$yl\_os := t \rightarrow \frac{1}{4}\pi\cos(t) \quad \textbf{(5.14)}$$

```
> plot([yl,yl_os],0..50,color=[black,red]);
```